
Animationen in Delphi

Grundlagen

Damit sich Objekte auf dem Bildschirm selbständig bewegen, benötigt man als erstes einen Timer, der in bestimmten Abständen das Neuzeichnen des Bildschirms auslöst. Obwohl reichlich ungenau und langsam, genügt für unsere Zwecke die Delphi-Komponente „Timer“ (auf der Komponentenseite „System“ zu finden). Mit Mausklick diese Kom​po​nente irgendwo auf dem Formular platzieren und im Objektinspektor die Eigenschaft „Interval“ auf ca. 10 (Millisekunden) einstellen. Ab jetzt wird etwa alle 10 Millisekunden das Ereignis „OnTimer“ ausgelöst. Im dazugehörenden Eventhandler (Doppelklick auf der Seite „Ereignisse“ neben „OnTimer“) schreiben wir die An​weisungen für die Animation.

Die eigentliche Animation besteht aus drei Schritten:

1)
Objekt an seiner letzten Position löschen, indem der Hintergrund darüber gezeichnet wird

2)
Objektkoordinaten für die nächste Position berechnen

3)
Objekt an neuer Position neu zeichen
Für die Bildschirmausgabe verwenden wir eine Paintbox (man könnte auch das Formular selbst dafür verwenden, dann bliebe aber kein Platz mehr für Schalter, etc.). Das Hintergrundbild und das zu bewegende Objekt (hier: ein Ballon) laden wir in je ein „Bitmap“-Objekt, von dort werden sie in jedem OnTimer-Ereignis mit der Methode „Draw“ nacheinander in die Paintbox kopiert.
Der Datentyp „TBitmap“

„TBitmap“ ist für Bilder gedacht und verfügt u.a. über die Methode „LoadFromFile“, die wir benötigen, um unsere Bilder zu laden. Um Bitmaps einsetzen zu können, sind drei Schritte notwendig:
a)
Deklaration einer (globalen) Variablen vom Typ „TBitmap“ (Für unser Beispiel deklarieren wir zwei Variablen: „Hg“ für das Hintergrundbild und „Ballon“ für die Grafik, die animiert werden soll):

var



Hg:TBitmap;



Ballon:TBitmap;
b)
Im Eventhandler „FormCreate“ (Ereignisse des Formulars, Doppelklick neben „OnCreate“) bekommen diese Variablen mit „Create“ Zugriff auf Methoden und Eigenschaften des TBitmap-Objekts:


procedure TForm1.FormCreate(Sender: TObject);


begin



Ballon:=TBitmap.Create;


Hg:=TBitmap.Create;
c)
Mit dem dritten Schritt wird uns (ebenfalls in „FormCreate“) endlich vom Betriebssystem der gewünschte Speicher für die Bilder reserviert, indem wir mit der Methode „LoadFromFile“ (verlangt in Klammer den genauen Dateinamen inklusive Erweiterung) die Grafik laden und dadurch soviel Speicher anfordern, wie die Grafik benötigt.
Grundsätzlich kann stattdessen einem Bitmap-Objekt auch Speicher zugewiesen werden, indem man Breite und Höhe angibt, aus denen sich der benötigte Speicherbedarf errechnet. In diesem Fall ist allerdings der reservierte Bereich leer:


Ballon.Width:=200;


Ballon.Height:=300;

Für unser Beispiel setzen wir bei beiden Bitmaps natürlich die Methode „LoadFromFile“ ein:

Hg.LoadFromFile('Hg.bmp');

Ballon.LoadFromFile('Ballon.bmp');

Ab jetzt können wir unsere Bilder mit der Methode „Draw“ in die Paintbox zeichnen. Allerdings soll der Ballon transparent vor dem Hintergrund erscheinen, daher ist (ebenfalls in „FormCreate“) noch seine Eigenschaft „Transparent“ auf „True“ zu setzen:
Ballon.Transparent:=True;

Die Animation

Wie erwähnt, befinden sich die Anweisungen für die eigentliche Animation im Eventhandler des Timers. Die Methode „Draw“ verlangt die Koordinaten, an denen das Bild gezeichnet werden soll, und den Namen des Objekts, das die zu zeichnende Grafik enthält. Da wir den Hintergrund über die gesamte Paintbox legen wollen, geben wir als Koordinaten „0,0“ (= linke obere Ecke der Paintbox) an. Die Koordinaten des Ballons dagegen berechnen wir in jedem Timer-Ereignis neu, indem (vorerst) einfach seine x-Koordinate um z.B. 5 Pixel erhöht wird (der Wert bestimmt die Geschwindigkeit der Animation). Dadurch wird er sich nach rechts bewegen. Natürlich benötigen wir zwei globale Variablen (in unserem Beispiel: „xAkt“ und „xAkt“), um uns die Koordinaten der letzten Position merken zu können, und müssen sie (wie üblich in „FormCreate“) mit einem Startwert initialisieren, z.B. mit „0“ (= ganz links am Rand der Paintbox):

var


xAkt, yAkt: integer

procedure TForm1.Timer1Timer(Sender: TObject);

begin


pbox1.Canvas.Draw(0,0,Hg);


xAkt:=xAkt+5;


pbox1.Canvas.Draw(xAkt,yAkt,Ballon);

end;
Die beiden Schalter, die auf dem Formular zu sehen sind, starten und beenden die Animation, indem der Timer ein- bzw. ausgeschaltet (d.h. seine Eigenschaft „Enabled“ auf „True“ oder „False“ gesetzt) wird:

procedure TForm1.bLosClick(Sender: TObject);

begin


Timer1.Enabled:=True;

end;

procedure TForm1.bStopClick(Sender: TObject);

begin


Timer1.Enabled:=False;

end;

[image: image9.png]Eigenschaften | Ereignisse

Enabled
Interval |30
Name  Timerl




Damit die Animation auch tatsächlich erst nach Klick auf den „Los“-Schalter startet, muss diese Eigenschaft des Timers (im Objektinspektor oder in „FormCreate“) auf „False“ voreingestellt werden. Den Hintergrund mit dem Ballon an seinem Startplatz wollen wir allerdings schon bei Programmstart sehen, nicht erst, wenn die Animation mit Klick auf den Button „Los“ in Gang kommt. Daher rufen wir unsere „Draw“-Anweisungen auch im Eventhandler des „Paint“-Ereignisses (wird bekanntlich bei Programmstart ausgelöst) der Paintbox auf (Doppelklick im Objektinspektor neben „OnPaint“. Selbstverständlich könnte man die Anweisungen, die nun in beiden Eventhandlern vorhanden sind, in eine eigene Prozedur auslagern):
procedure TForm1.pbox1Paint(Sender: TObject);

begin


pbox1.Canvas.Draw(0,0,Hg);


pbox1.Canvas.Draw(xAkt,yAkt,Ballon);

end;
Flimmerfreie Animation („Buffering“)
Zwar wandert unser Ballon brav über den Bildschirm, aber, wie man sieht, nicht eben flimmerfrei. Dieser unangenehme Effekt entsteht durch das ständige Übereinanderzeichnen von Hintergrund und Ballon. In der Praxis verlegt man daher die „Draw“-Anweisungen in einen unsichtbaren Bereich und kopiert erst dann, wenn sie abgeschlossen sind, das fertige Bild in die Paintbox. Diese „Buffering“ genannte Technik benötigt ein weiteres Bitmap-Objekt als „Puffer“. Wir erzeugen es wie die beiden anderen in „FormCreate“, allerdings ohne die Methode „LoadFromFile“ aufzurufen. Stattdessen reservieren wir uns genauso viel Speicher, wie das Hintergrundbild benötigt, indem die Eigenschaften „Width“ und „Height“ auf die gleichen Werte gesetzt werden (vgl.S.2):
Puffer:=TBitmap.Create;

Puffer.Width:=Hg.Width;

Puffer.height:=Hg.Height;

Hintergrund und Ballon werden nun zuerst in diesen Puffer gezeichnet und anschließend der gesamte Pufferinhalt in die Paintbox:


Puffer.Canvas.Draw(0,0,Hg); {Hintergrund in Puffer zeichnen}

Puffer.Canvas.Draw(xAkt,yAkt,Ballon); {Ballon in Puffer zeichnen}

pbox1.Canvas.Draw(0,0,Puffer); {Puffer in Paintbox kopieren}
Spätestens jetzt zahlt sich eine eigene Prozedur dafür aus (die wir am besten in die Klassendeklaration unseres Formulars – nach „type“ – einbinden):

type

TForm1 = class(TForm)

Timer1: TTimer;

pbox1: TPaintBox;

bLos: TButton;

bStop: TButton;

procedure FormCreate(Sender: TObject);

procedure Timer1Timer(Sender: TObject);

procedure bLosClick(Sender: TObject);

procedure bStopClick(Sender: TObject);

procedure pbox1Paint(Sender: TObject);

procedure Zeichnen;


private


{ Private-Deklarationen }


public


{ Public-Deklarationen }


end;

var

Form1: TForm1;

xAkt,yAkt:integer;

Hg:TBitmap;

Ballon:TBitmap;

Puffer:TBitmap;

( ... )

procedure TForm1.FormCreate(Sender: TObject);

begin


hg:=TBitmap.Create;


hg.LoadFromFile('hg.bmp');


Ballon:=TBitmap.Create;


Ballon.LoadFromFile('ballon.bmp');


Ballon.Transparent:=True;


Puffer:=TBitmap.Create;


Puffer.Width:=Hg.Width;


Puffer.Height:=Hg.Height;


xAkt:=0;


yAkt:=0;

end;

procedure TForm1.Zeichnen;

begin


Puffer.Canvas.Draw(0,0,Hg);


Puffer.Canvas.Draw(xAkt,yAkt,Ballon);


pbox1.Canvas.Draw(0,0,Puffer);

end;

procedure TForm1.pbox1Paint(Sender: TObject);

begin


Zeichnen;

end;

procedure TForm1.Timer1Timer(Sender: TObject);

begin


xAkt:=xAkt+5;


Zeichnen;

end;

Bewegen in beliebige Richtung

Unser Ballon fliegt momentan ausschließlich waagrecht dahin. Um ihn an beliebige Stelle schicken zu können, werden wir auf einen Mausklick warten und ihn anschließend an die Stelle bewegen, an die geklickt wurde (die beiden Schalter „Los“ und „Stop“ benötigen wir damit nicht mehr). Für die Auswertung von Mausklicks ist zwar der Eventhandler zum „OnClick“-Ereignis zuständig. Dummerweise liefert er aber nicht die aktuelle Mausposition mit. Daher müssen wir stattdessen auf das Ereignis „OnMausUp“ reagieren, das in den Parametern „x“ und „y“ die Koordinaten der Maus verrät:
procedure TForm1.pbox1MouseUp(Sender: TObject; Button: TMouseButton;

  Shift: TShiftState; X, Y: Integer);

begin


(...)

end;

Für die Neuberechnung der Koordinaten müssen wir nun drei Werte ermitteln:

· Die Länge der Strecke, die das Objekt vom Start bis zum Ziel zurücklegt

· Die Anzahl der Timer-Events, bis es am Ziel ankommt

· Die Summanden, um die jeweils x- und y-Koordinate des Objekts pro Timerintervall erhöht werden

Wir haben bereits zwei Variablen („xAkt“ und „yAkt“ – s.S.3) für die aktuelle Position des Ballons angelegt. Somit besitzen wir Startpunkt (xAkt|yAkt) und Zielpunkt (= die Stelle, an die geklickt wurde). Nehmen wir an, der Startpunkt sei (16|12) gewesen und es wurde an die Stelle (72|48) geklickt. Unser Objekt muss demnach auf der x-Achse (nenne wir sie „StreckeX“) 72-16=56 Pixel zurücklegen, auf der y-Achse („StreckeY“) 48-12=36 Pixel. Die Länge der direkten Verbindung zwischen Start- und Zielpunkt (nennen wir sie „StreckeC“) lässt sich nun mit Hilfe des pythagoreischen Lehrsatzes errechnen:

 Pixel. Zum Quadrieren gibt´s die Funktion „sqr“ (= „square“), zum Wurzelziehen die Funktion „sqrt“ (= square root“):

StreckeX:=x-xAkt;


StreckeY:=y-yAkt;


StreckeC:=sqrt(sqr(StreckeX)+sqr(StreckeY));

Die Geschwindigkeit, mit der sich der Ballon bewegt, wird durch die Anzahl Pixel bestimmt, die er in jedem Timer-Ereignis vorrücken darf. Wir haben dafür 5 Pixel pro Durchgang festgelegt (s.S.2). Daraus berechnen wir die Anzahl der Timer-Events (= die Anzahl der Durchgänge), die durchlaufen werden, bis der Ballon das Ziel erreicht hat:
AnzDurchgange:=round(StreckeC/Geschwindigkeit);
Ergibt 66,57 / 5 = 13,31. Wir runden auf 13 und können damit endlich bestimmen, um wieviel die x- und y-Koordinaten des Ballons in jedem Durchgang erhöht werden müssen (also den Summand), damit er sich möglichst genau entlang der „StreckeC“ bewegt:
SummandX:=StreckeX/AnzDurchgange;

SummandY:=StreckeY/AnzDurchgange;

„SummandX“ ergibt in unserem Beispiel 4,31, „SummandY“ ergibt 2,77. Natürlich können wir der Draw-Methode nur Ganzzahlen übergeben (der Bildschirm besteht nun einmal aus ganzen Bildpunkten) und runden deshalb diese Werte vor der Übergabe. Merken und aufaddieren müssen wir aber unbedingt die reellen Zahlen, um Rundungsfehler zu vermeiden, die unseren Ballon ganz woanders hinführen würden. Wir arbeiten also mit zwei unterschiedlichen Variablen: den integer-Variablen „xAkt“ und „yAkt“, die der Draw-Methode übergeben werden, und den beiden extended-Variablen „xAktExt“ und „yAktExt“, die nach jedem Durchgang um die oben errechneten Summanden „SummandX“ und „SummandY“ erhöht werden:

xAktExt:=xAktExt+SummandX;

yAktExt:=yAktExt+SummandY;

xAkt:=round(xAktExt);

yAkt:=round(yAktExt);

(...)

Puffer.Canvas.Draw(xAkt,yAkt,Ballon);
In der folgenden Abbildung sind alle 13 Positionen des animierten Ballons mit den jeweiligen x- und y-Koordinaten (zur besseren Übersicht auf eine Kommastelle gerundet) der einzelnen Durchgänge eingetragen:

Die globale Variable „AnzDurchgange“, in der die Anzahl der Timer-Events eingetragen wurde, die der Ballon vom Start bis ins Ziel benötigt, verwenden wir im Timer-Eventhandler als Zähler, der nach jedem Durchgang um eins verringert wird (mit der Funktion „dec“. Ebensogut hätte man natürlich schreiben können: AnzDurchgange:= AnzDurchgange‑1). Ist er bei 0 angelangt, stoppen wir die Animation, indem wir den Timer ausschalten:

procedure TForm1.Timer1Timer(Sender: TObject);

begin


if AnzDurchgange > 0 then begin


xAktExt:=xAktExt+SummandX;



yAktExt:=yAktExt+SummandY;



xAkt:=round(xAktExt);



yAkt:=round(yAktExt);



Zeichnen;



dec(AnzDurchgange);


end else Timer1.Enabled:=False;

end;

Hier die wichtigsten Teile des gesamten Quelltexts:

var


Form1: TForm1;


xAkt,yAkt,Geschwindigkeit,AnzDurchgange:integer;


Hg:TBitmap;


Ballon:TBitmap;


Puffer:TBitmap;


SummandX,SummandY,xAktExt,yAktExt:extended;

procedure TForm1.FormCreate(Sender: TObject);

begin


hg:=TBitmap.Create;


hg.LoadFromFile('hg.bmp');


Ballon:=TBitmap.Create;


Ballon.LoadFromFile('ballon.bmp');


Ballon.Transparent:=True;


Puffer:=TBitmap.Create;


Puffer.Width:=Hg.Width;


Puffer.Height:=Hg.Height;


xAkt:=0;


yAkt:=0;


Geschwindigkeit:=5;

end;

procedure TForm1.Zeichnen;

begin


Puffer.Canvas.Draw(0,0,Hg);


Puffer.Canvas.Draw(xAkt,yAkt,Ballon);


pbox1.Canvas.Draw(0,0,Puffer);

end;

procedure TForm1.Timer1Timer(Sender: TObject);

begin


if AnzDurchgange > 0 then begin



xAktExt:=xAktExt+SummandX;



yAktExt:=yAktExt+SummandY;



xAkt:=round(xAktExt);



yAkt:=round(yAktExt);



Zeichnen;



dec(AnzDurchgange);


end else Timer1.Enabled:=False;

end;

procedure TForm1.Berechnen(x,y:integer);

var


StreckeX,StreckeY:integer;


StreckeC:extended;

begin


StreckeX:=x-xAkt;


StreckeY:=y-yAkt;


StreckeC:=sqrt(sqr(StreckeX)+sqr(StreckeY));


AnzDurchgange:=round(StreckeC/Geschwindigkeit);


SummandX:=StreckeX/AnzDurchgange;


SummandY:=StreckeY/AnzDurchgange;


xAktExt:=xAkt;


yAktExt:=yAkt;

end;

procedure TForm1.pbox1MouseUp(Sender: TObject; Button: TMouseButton;

  Shift: TShiftState; X, Y: Integer);

begin


Timer1.Enabled:=True;


Berechnen(x,y);


Zeichnen;

end;

Wer will, kann die Koordinaten in der Draw-Methode so korrigieren, dass sich nicht die linke obere Ecke, sondern die Mitte des Ballons zur Mausposition bewegt – sieht natürlich besser aus:


Puffer.Canvas.Draw(xAkt-(Ballon.Width div 2),yAkt-(Ballon.Height div 2),Ballon);

